
Mise en bouche <

Au fil du temps, le langage SAS s'est enrichi de mécanismes divers comme le
langage macro, de nouvelles fonctions, de nouvelles routines, de nouvelles
procédures.

De nouvelles composantes se sont également ajoutées le tout dans le but de
simplifier I'accés aux données et la production de résultats.

Il n'en demeure pas moins que parmi toutes les possibilités offertes, certaines
sont demeurées un peu plus dans I'ombre que d'autres.

Ainsi en est-il de la routine CALL EXECUTE apparue dans la version 6.07 en
avril 1991.

CALL EXECUTE est a prime abord une routine de I'étape DATA qui interagit

avec le dispositif macro comme le font par exemple le CALL SYMPUTX et la
fonction RESOLVE.

Peut-t'on en tirer des bénéfices dans nos traitements?

C'est a vous de découvrir la réponse a cette question.



CALL EXECUTE en bref

Exemple a partir de la bibliotheque SASHELP:
* Sans CALL EXECUTE il faudrait coder ceci:

e
14

proc print data=sashelp.class;
run;

proc print data=sashelp.cars;
run;

proc print data=sashelp.heart;
run;

* Avec CALL EXECUTE ;
* Générer des PROC PRINT pour chacune des tables ;

data null ;
do table = 'class", '"cars", "heart";
call execute("proc print data=sashelp." ||table|]| ";");
call execute("run;");
end;

run,




CALL EXECUTE en bref

Extrait du journal (log) résultant:

Le code SAS génére est exécuté apres I'étape DATA.
Les + témoignent du
code genére comme
pour un %include

NOTE: CALL EXECUTE generated line.
1 + proc print data=sashelp.class;
2 + run;

NOTE: There were 19 observations read from the data set
SASHELP.CLASS.
NOTE: PROCEDURE PRINT a utilisé (Durée totale du traitement)

real time 0.27 seconds
cpu time 0.04 seconds
3 + proc print data=sashelp.cars ;
4 + run;

NOTE: There were 428 observations read from the data set
SASHELP.CARS.
NOTE: PROCEDURE PRINT a utilisé (Durée totale du traitement)

etc...




Alternatives, complements,
autres pistes

Exemple d'exécution conditionnelle d'un énoncé a portée globale avec le
%IF %THEN %DO en "open code":

* Arrét du traitement si table wvide ;
* Version macro avec %if en open code (SAS 9.4M5) ;

%let dset = TableVide;
%let dsid = %sysfunc (open (&dset)) ;
%let nobs = %sysfunc(attrn(&dsid,NOBS)) ;
%if &nobs=0 %$then %do;
%$let rc = $sysfunc(close(&dsid)) ;
sput NOTE: &=dset &=dsid &=nobs &=rc;
endsas;
%$end;

Dans le journal (log):

NOTE: DSET=TableVide DSID=1 NOBS=0 RC=0
Error: La connexion au serveur a été perdue. (Enterprise Guide)
Avec le Display Manager, la session se termine immmédiatement




Langage %macro VS étape DATA

Langage macro || Etape DATA

%LET === CALL SYMPUTX

%MACRO &= CALL EXECUTE
%IF %THEN %DO

%SYSFUNC rL{Prect{[rr Fonction(x)




Import / Export Excel

Exemple:

* Allouer la bibliotheque (classeur)Excel ;
libname MONEXCEL xlsx "C:\temp\SAS\excel\CLASSEUR1l.xlsx";

* Générer des étapes DATA pour créer des tables SAS ;

data null ;
set sashelp.vmember (where=(libname="MONEXCEL")) ;

call execute ( "data work."| |memname]| |"; ") ;

call execute ( " set MONEXCEL."| |memname| |";" ),

call execute ( "run; "y
run;

C1 signifie Classeur1 et Fx signifie Feuille x

CLASSEUR1 .xIsx contient 2 feuilles: C1F1 et C1F2

& libname ¢ memname ¢ memtype ¢ dbms_memtype & engine & index & path

1 [MONEXCEL |C1F1 DATA TABLE XLSX no C:\temp\sas\excel\CLASSEUR1.xlsx

2 MONEXCEL C1F2 DATA TABLE XLSX no C:\temp\sas\excel\CLASSEUR1.xlsx




Import / Export Excel

Extrait du journal (log) résultant:

1

A B
NOTE: CALL EXECUTE generated line
& C1F1_ColA & C1F1_ColB
+ data work.C1F1 ; e cr S 1 | C1F1_Cell_A2 | C1F1_Cell_B2
+ set MONEXCEL.C1F1 ; 2 CI1F1_Cell_A2 C1F1_Cell_B2 el el

2
3

NOTE :
NOTE :
NOTE :
NOTE :

o 01

NOTE :
NOTE :
NOTE :

+ run; 3 CI1F1_Cell_A3 C1F1_Cell_B3 2 CI1F1_Cell_A3 C1F1_Cell_B3

The import data set has 2 observations and 2 variables.

There were 2 observations read from the data set MONEXCEL.C1lF1l.
The data set WORK.ClF1l has 2 observations and 2 variables.
DATA statement a utilisé (Durée totale du traitement)

real time 0.02 seconds
cpu time 0.03 seconds
A B
+ data work.ClF2 ; 1 [C1F2 ColA  C1F2 ColB & C1F2.ColA & C1F2_ColB

+ set MONEXCEL.C1lF2 ;

2 C1F2_Cell_ A2 C1F2_Cell B2 || CIF2Cell A2 C1F2_Cell B2
+ run; I I

3 |C1F2 Cell A3 C1F2 Cell B3 ||2 CiF2.Cel A3 C1F2_Cell_B3

The import data set has 2 observations and 2 variables.
There were 2 observations read from the data set MONEXCEL.Cl1lF2.
The data set WORK.C1lF2 has 2 observations and 2 wvariables.




Import / Export Excel

Exemple:

* Exporter plusieurs tables SAS dans un classeur Excel;
$let mypath=C:\temp\SAS\excel;

data null ;
set sashelp.vmember (

where=(libname="SASHELP" and memname in ("CLASS" "CARS"

"RETAIL")) );

call execute( "proc export data="||strip(libname) ||"."||strip (memname)) ;

call execute( ' outfile="&mypath\CombiHelp.xlsx"' ) ;

call execute( " dbms=xlsx replace;" ) ;

call execute( " sheet="||quote(strip(memname)) ||" ;" ) ;
)

call execute( "run;"

run,
SASHELP.VMEMBER:
Obs. | libname memname memtype nobs nvar
1 | SASHELP @ CARS DATA 428
2 | SASHELP @ CLASS DATA 19
3 | SASHELP @ RETAIL DATA 58

15

Ne

Import / Export Excel

Extrait du journal (log) résultant:

NOTE :

b WdhR

NOTE :
NOTE :
NOTE :

©O© 00 Jd o

10
NOTE :
NOTE :

CALL EXECUTE generated line.

+ proc export data=SASHELP.CARS
+ outfile="C:\temp\SAS\excel\CombiHelp.xlsx"
+ dbms=xl1sx replace;
+ sheet="CARS";
+ run;

The export data set has 428 observations and 15 variables.

"C:\temp\SAS\excel\combihelp.xlsx" file was successfully created.

PROCEDURE EXPORT a utilisé (Durée totale du traitement)
real time 1.97 seconds
cpu time 0.03 seconds

+ proc export data=SASHELP.CLASS
+ outfile="C:\temp\SAS\excel\CombiHelp.x1lsx"
+ dbms=xl1sx replace;
+ sheet="CLASS";
+ run;
The export data set has 19 observations and 5 variables.

"C:\temp\SAS\excel\combihelp.xlsx" file was successfully created.

.etc.

Fichier texte auto-documenteé

Scénario:

Un fichier texte contient des lignes de données qui sont précédées de lignes
décrivant ces mémes données.

En fait un tel fichier texte ressemble conceptuellement a une table SAS ou a
une table de base de données relationelle. Dans ce genre de structure on
retrouve les données elles-mémes et l'information décrivant ces données.

Comment faire avec une étape DATA et a 'aide de la routine CALL EXECUTE

pour lire un type de fichier texte contenant la description des données suivie
des données elles-mémes et le transformer en table SAS?

SAS Data Set

descriptor
_— portion

- o portion

Fichier texte auto-documenteé

Exemple de fichier texte:

En colonne 1 le "V" signifie description de Variable contenant le nom de la
variable et le format de lecture associé. Un commentaire peut
optionnellement suivre.

Toujours en colonne 1, le "D" identifie une ligne de Données qui sera lue selon
les modalités définies par les lignes "V" précédentes.

Les autres lignes sont ignorées — considerees comme des commentaires.

—==—t--=-l---—t--=-2--——f--=-3--——f--——4--——f--—-5--——4--- N
* Spécifications des Variables et des Formats: 1
* Nom........ Format. .Commentaire 2
V NMVAR 3. Variable numérique 3
V DTVAR YYMMDD10. Variable de date 4
V CHVAR SCHAR4. Variable de type caractere 5
* Les données suivent selon le format de lecture ci-haut: 6
D 148,1994-07-15,ABCD < 7
D 295,1993-12-12,WXYZ 8
D 321,2021-06-22 ,MNOP 9

13



Fichier texte auto-documenteé

On aimerait retrouver du code SAS ressemblant a:

data TabOut;
infile inVD dsd truncover fistobs=7;

input @3 * Nom. ..
NMVAR : 3. V NMVAR
DTVAR : YYMMDDI1O0. < || Vv DTVAR
CHVAR : SCHARA. V CHVAR
run,

infile inVD .... firstobs=7 obs. | NMVAR
D 148,1994-07-15,ABCD <== 1 148
D 295,1993-12-12, WXYZ E—) 2| 295
D 321,2021-06-22,MNOP 3| 321

DTVAR
1994-07-15
1993-12-12
2021-06-22

Table WORK.TABOUT

CHVAR
ABCD
WXYZ
MNOP

14



Fichier texte auto-documenteé

Code SAS proposé — partie 1:

Génération des énoncés préalables pour créer une étape DATA

data null_ ;

length VarLu $ 32 FmtLu $ 32;
* VarLu pour Variable Lue et FmtLu pour Format Lu;

if n =1 then
call
call
call
end;

infile inVD;
input RecType

do;

execute ("data TabOut; ")
execute (" infile inVD dsd truncover;");
execute (" input @3 ")
Supcasel. @;

15



Fichier texte auto-documenteé

Code SAS proposé — partie 2:

Génération des €lements pour lire les variables et réajustement du INFILE afin
de "pointer" sur la premiére ligne de données "D" avec FIRSTOBS=.

if RecType = "V" then
do; /* On va lire toutes les définitions de wvariables */
input VarLu $ FmtLu $;

call execute (" "|| catx(" ", VarLu, ":", FmtLu)) ;
end;
else
if RecType = "D" then
do; /* Trouver la premiére ligne seulement */
call execute (" ; ")
call execute(" infile inVD firstobs="||put( n ,z3.)[]|";")
call execute("run; ")
stop; /* On arréte aussitdét */
end;
run ; Remarquer ceci ]

.
14

.
14

.
14

Fichier texte auto-documenteé

Code SAS finalement généré:

NOTE: CALL EXECUTE generated line.

1 + data TabOut;

2 + infile inVD dsd truncover; . :

3 + input @3 Premiere Ilgne]
4 + NMVAR : 3. de données

5 + DTVAR : YYMMDD1O.

6 + CHVAR : S$SCHARA.

7 + ;

8 + infile inVD firstobs=007;

9 + run;

Rappel - le fichier texte a lire:

—=——4-—--l--—-4-—--2--——4-—--3--——4-—-—4--——4-—--5----4--- 1
Spécifications des Variables et des Formats:

Nom........ Format..Commentaire

NMVAR 3. Variable numérique

DTVAR YYMMDD10. Variable de date

CHVAR SCHAR4. Variable de type caracteére

Les données suivent selon le format de lecture ci-haut:
148,1994-07-15,ABCD <
295,1993-12-12,WXYZ
321,2021-06-22,MNOP

DODUO x4 << * %
WoOJOU B WN Hl

firstobs=




Tracer PROC REPORT

Scénario:

La procédure REPORT est riche de fonctionnalités et de possibilités.
Cependant il est parfois difficile de bien comprendre ce qui se passe
lorsqu'on construit un traitement avec cette procedure notamment avec
I'utilisation des blocs COMPUTE.

Quand on veux déboguer on dispose d'options comme LIST et SHOWALL
mais on n'a pas de mécanisme de tragage comme on peut en retrouver avec
I'option DEBUG de I'étape DATA. Par contre, I'option OUT nous permet
d'obtenir une table SAS gu'on peux examiner pour nous aider a mieux
comprendre ce qui se passe.

De plus, nous pouvons aussi utiliser la routine CALL EXECUTE d'une maniére
orignale et inusitée dans les blocs COMPUTE et ainsi disposer d'un outil
d'analyse dans notre quéte du resultat recherche.

Cette approche n'est pas une panacée et n'est sans doute pas la premiére
approche a utiliser quand on fait la mise au point d'un traitement PROC
REPORT mais elle ouvre un sentier qui mérite d'étre exploreé.

Tracer PROC REPORT

Quelques considérations concernant les blocs COMPUTE:

Type de bloc COMPUTE vs énoncés IF de I'étape DATA:

COMPUTE BEFORE IF _N_=1
COMPUTE AFTER IF FIN (SET EOF=FIN)

COMPUTE BEFORE variable IF FIRST.variable
COMPUTE AFTER variable IF LAST.variable

Ordre d'exécution des blocs compute en général:

1. COMPUTE item-de-rapport;
2. COMPUTE BEFORE;

3. COMPUTE BEFORE cible;

4. COMPUTE BEFORE PAGE_;
5

6

7

. COMPUTE AFTER;
. COMPUTE AFTER cible;
. COMPUTE AFTER _PAGE_;

19



Tracer PROC REPORT

Code SAS proposé — partie 1:

Début de la procédure REPORT — les colonnes et leurs définitions:

proc report data=Eleves missing
/* list showall */
out=RptEleves;

columns Genre Nom Age Age=AgeMoy;
define Genre / order;

define Nom / display;

define Age / display;

define AgeMoy / mean noprint;

...Suite prochaine diapo

Obs.

-~ o 0 A W N =

Table Eleves

Nom Genre Age
Claude M 17
Elodie F 13
Samuel X 18
Francois M 12
Phong F 14
Tiago M 15
Mila F 16

20



Tracer PROC REPORT

Code SAS proposé — partie 2:

Suite et fin de la procédure REPORT - les blocs COMPUTE:

compute after;
call execute('%put NOTE: Compute After;');
$EcrireValeurs() ;

endcomp ;

compute after Genre;
call execute('%put NOTE: Compute After Genre;');
$EcrireValeurs() ;

endcomp ;

compute Age;
call execute('%put NOTE: Compute Age;');
$EcrireValeurs() ;
endcomp ;
run;

21



Tracer PROC REPORT

Code SAS proposé — partie 3:
Définition d'une macro utilitaire qui va générer l'information de tracage:

$macro EcrireValeurs () ;
call execute (
cat('%put INFO: > '

, ' _BREAK=' , substr( break ,1,10)
, ' Genre=' , Genre

, ' Nom=' , Nom

, ' Age=' , Age

, ' AgeMoy=' , AgeMoy

1
V4 ’

) ) ;

$mend EcrireValeurs;

Bien que présentée ici, la macro doit bien sir étre placée avant la procédure REPORT.
On verra plus loin que la macro va générer dans le journal (log) des messages comme
celui-ci:

INFO: > _BREAK= Genre=F Nom=Elodie Age=13 AgeMoy=.

22



Tracer PROC REPORT

Résultat produit:

Rappel des enonceés de la procedure:

proc report data=Eleves missing out=RptEleves;

columns Genre Nom Age Age=AgeMoy;

define Genre / order;

define Nom / display;

define Age / display;

define AgeMoy / mean noprint;

compute after;
call execute('%put NOTE: Compute After;');
$EcrireValeurs() ;

endcomp ;
etc....

compute Age;

call execute('%put NOTE: Compute Age;')
$EcrireValeurs() ;

endcomp ;

run;

Rapport produit

Genre Nom

F Elodie
Mila
Phong

M Claude
Francois
Tiago

X Samuel

Age
13
16
14
17
12
15
18

23



Tracer PROC REPORT

Extrait du journal SAS — 90 lignhes au total :

NOTE :
INFO:
NOTE :
INFO:
NOTE :
INFO:
NOTE :
INFO:
NOTE :
INFO:

INFO:
NOTE:
INFO:
NOTE:
INFO:
NOTE:
INFO:
NOTE:
INFO:

Compute
>
Compute
>
Compute
>
Compute
>
Compute

Compute
>
Compute
>
Compute
>
Compute
>

Age

_BREAK=

Age

_BREAK=

Age

_BREAK=

Age

BREAK=Genre

Afﬁgr Genre

_BREAK=Genre

_BREAK=

Age

BREAK=Genre

Aftgr Genre

_BREAK=Genre

Age

BREAK= RBREAK _

Aftgr

_BREAK= RBREAK _

Genre=F
Genre=
Genre=
Genre=F

Genre=F

Genre=i
Genre=X
Genre=X
Genre=

Genre=

Nom=Elodie
Nom=Mila
Nom=Phong
Nom=

Nom=
N;m;Saﬁuél
Nom=

Nom=

Nom=

Nom=

Age=13 AgeMoy=.
Age=16 AgeMoy=.
Age=14 AgeMoy=.
Age=. AgeMoy=.
Age=.

Age=18 AgeMoy=.

Age=. AgeMoy=.
Age=. AgeMoy=18
Age=. AgeMoy=.
Age=. AgeMoy=15

AgeMoy=14.33

24



Tracer PROC REPORT

Fichier RptEleves produit par PROC REPORT

Rappel des eénonceés de la procédure

proc report data=Eleves missing out=RptEleves;
columns Genre Nom Age Age=AgeMoy:’

define Genre / order;
define Nom / display;
define Age / display;
define AgeMoy / mean noprint;
compute after;
call execute('%put NOTE: Compy
$EcrireValeurs() ;
endcomp ;
endcomp ;
compute Age;
call execute('%put NOTE: Compy
$EcrireValeurs() ;
endcomp ;
run;

Obs.

O 0 ~N S AW N =

—
[ S s |

Fichier créé par la procédure

Genre | Nom
Elodie
Mila
Phong

Claude
Francois

Tiago

Samuel

X X = = Z= . = mMm m M| m

Age

13
16
14

17
12

18

AgeMoy

13.0000
16.0000
14.0000
14.3333
17.0000
12.0000
15.0000
14.6667
18.0000
18.0000
15.0000

_BREAK_

Genre

Genre

Genre
RBREAK

Alternatives, complements,
autres pistes

Dans les deux premiers exemples on a utilisé le CALL EXECUTE pour générer
du code SAS. La génération de code est une pratique qu'on retrouve de
différentes facons dans beaucoup de traitements SAS.

Le langage macro est sans doute la maniere la plus courante de créer du code
SAS car c'est son role fondamental.

La création d'énoncés SAS dans un fichier avec une étape DATA et qu'on
réecupére avec un %INCLUDE est une autre fagcon de faire.

Plus récemment (SAS9.4M5), I'utilisation en "open code" du %IF %THEN
%ELSE a ouvert une porte pour faciliter la génération/modification de code
sans avoir a encapsuler le tout dans le duo %MACRO + %MEND.

La fonction DOSUBL apparue a la version 9.3 a poussé un peu plus loin la
possibilité d'exécuter du code SAS sans attendre la fin de I'étape DATA
comme c'est le cas pour CALL EXECUTE.

26



Alternatives, complements,
autres pistes

Exemple de génération de code avec une combinaison d'étape DATA et du
%INCLUDE:

* Fichier texte temporaire pour accueillir le code SAS généré;
filename CodeSAS temp;

* Génération du code SAS;
data null ;
set TablesChoisies;
file CodeSAS;
put "data " NomTable ";" ;

put " set sashelp." NomTable ";";
put "run;";
run;

%$include CodeSAS /source2;

27



Alternatives, complements,
autres pistes

Résultat abrégé dans le journal (log):

NOTE: The file CODESAS is:
Nom du fichier=C:\Users\...\SAS Temporary...,
NOTE: 9 records were written to the file CODESAS.
NOTE: There were 3 observations read from the data set
WORK .TABLESCHOISIES.
NOTE: DATA statement a utilisé (Durée totale du traitement)

62 $include CodeSAS /source2;

NOTE: $%INCLUDE (level 1) file CODESAS is file C:\Users...

63 +data class ;

64 + set sashelp.class ;

65 +run;

NOTE: There were 19 observations read from ... SASHELP.CLASS.

NOTE: The data set WORK.CLASS has 19 observations and 5 variables
etc.

28



Alternatives, complements,
autres pistes

Clin d'oeil a la fonction DOSUBL (V9.3 en 2011) vs CALL EXECUTE:

* Exécution du code en différeé;
data null ;
putlog "AAA";
call execute('data null ; ")
call execute(' putlog "BBB";');
call execute('run; ")
putlog "CCC";
run;

4

* Exécution du code dynamiquement;
data null ;

putlog "AAA";

rc = dosubl('data null ; '

' putlog "BBB"; '

putlog "CCC";
run;

4

|
|
'run; ' ) .

Résultat dans le journal (log):

AAA
CCC
BBB

Résultat dans le journal (log):

AAA
BBB
CCC

29



Approche a éviter @

Acces répété a la méme source de données:

data null ;
set SASHELP.CARS;
by MAKE;
if first.MAKE then
do;

call execute('titlel "Manufacturier '||trim(MAKE) ||'";"')

call execute('proc print data=SASHELP.CARS
(where=(MAKE=""| |trim(MAKE) | |'")) ;") ;
call execute(' wvar MODEL TYPE;');
call execute('run;');
end;
run;

Approche de génération de code qui peut poser probleme pour de

volumineuses tables. Lecture répétée de la méme table n'est généralement

pas souhaitable quelque soit la méthode employée (étape DATA, macro,
CALL EXECUTE).

30



Conclusion

CALL EXECUTE: un outil pour générer du code SAS

Pour compléter son coffre d'outils en la matiere

Pas une panacée ou un outil universel

Simple d'utilisation

Accessible pour les personnes moins familiéres avec le langage macro
Peut étre utile dans certaines circonstances

Usage non trivial avec la procédure REPORT (résultat pas évident !)

Questions ?
Commentaires ?
Suggestions ?

Générer du code SAS’
avec
CALL EXECUTE

Contact:

Y, B — M

2 courriel=lowcase (cat (reverse ("Alain") ,reverse ("-") , "_")) ;
3 putlog "NOTE: " courriel=;

4 run;

NOTE : courriel=nia1a_

Marques de commerce

SAS et tous les autres noms de produits et services de SAS Institute Inc. sont des marques de commerce de SAS Institute Inc. aux Etats-Unis et
dans les autres pays. ® indique l'inscription aux Etats-Unis. Les autres noms de produits ou marques sont des marques déposées ou des marques

de commerce de leurs compagnies respectives.

/*

----------------------------------------------------

   Générer du code SAS® avec CALL EXECUTE 

   Alain Voyer, Club des utilisateurs SAS de Québec 

   http://wp.clubsasquebec.ca/

   3 février 2022 

   nialareyov@gmail.com

----------------------------------------------------

*/



options msglevel=i;



*----------------------;

* CALL EXECUTE en bref ;

*----------------------;



* Sans CALL EXECUTE il faudrait coder ceci: ;



 proc print data=sashelp.class;

 run;



 proc print data=sashelp.cars;

 run;



 proc print data=sashelp.heart;

 run;



 * Avec CALL EXECUTE ;

 * Générer des PROC PRINT pour chacune des tables ;



 data _null_ ;

   do table = "class", "cars", "heart";

      call execute("proc print data=sashelp." ||table|| ";");

      call execute("run;");

   end;

 run;





*-----------------------------------------------------------------;

* Exemple d'exécution conditionnelle d'un énoncé à portée globale ;

*-----------------------------------------------------------------;



* Création d'une table vide ;

data TableVide;

  if 0 then set sashelp.class;

  stop;

run;



* Arrêt du traitement si table vide ;

data _null_;

  if 0 then set TableVide nobs=NbrObs;

  if NbrObs = 0 then do;

       putlog "NOTE: Fin volontaire du programme - Table vide";

       call execute("endsas;");

     end;

  stop;

run;



* Version macro avec %if en open code (SAS 9.4M5) ;



 %let dset = TableVide;

 %let dsid = %sysfunc(open(&dset));

 %let nobs = %sysfunc(attrn(&dsid,NOBS));

 %if &nobs=0 %then %do;

         %let rc = %sysfunc(close(&dsid));

         %put NOTE: &=dset  &=dsid  &=nobs  &=rc;

         endsas;     

     %end;





*----------------------------------------------;

* Importation et exportation de classeur EXCEL ;

*----------------------------------------------;



%let mypath=C:\temp\sas\excel; /* À changer au besoin */



* Allouer la bibliothèque (classeur) Excel ;

libname MONEXCEL xlsx "&mypath\CLASSEUR1.xlsx";



* Générer des étapes DATA pour créer des tables SAS ;



data _null_;

  set sashelp.vmember (where=(libname="MONEXCEL"));

  call execute ( "data work."||memname||";     " );

  call execute ( "  set MONEXCEL."||memname||";" );

  call execute ( "run;                         " );

run;





* Exporter plusieurs tables SAS dans un classeur Excel ;



%let mypath=C:\temp\sas\excel; /* À changer au besoin */



data _null_;

  set sashelp.vmember(

      where=(libname="SASHELP" and memname in("CLASS" "CARS" "RETAIL")));

  call execute( "proc export data="||strip(libname)||"."||strip(memname) );

  call execute( '     outfile="&mypath\combihelp.xlsx"'                  );

  call execute( "     dbms=xlsx replace;"                                );

  call execute( "  sheet="||quote(strip(memname))||";"                   );

  call execute( "run;");

run;



*----------------------------------;

* Fichier paramètre auto-documenté ;

*----------------------------------;



* Création du fichier paramètre contenant les noms et

  formats de variables ainsi que les données.

  La colonne 1 indique le choix entre variable (V) et donnée (D).

  Pour les variables on retrouve le nom à partir de la colonne 3

  suivi du nom de format de lecture associé.

  On commence par définir les variables (V) et ensuite les données (D).

  Les données suivent les indications définies par les variables.

;



* Création du fichier texte auto-documenté ;

filename inVD temp lrecl=80;



data _null_; 

  infile datalines;

  input;

  file inVD;

  put _infile_;

  list;

datalines;

* Spécifications des Variables et des Formats:

* Nom........Format..Commentaire

V NMVAR          3.  Variable numérique

V DTVAR   YYMMDD10.  Variable de date

V CHVAR     $CHAR4.  Variable de type caractère

* Les données suivent selon le format de lecture ci-haut: 

D 148,1994-07-15,ABCD

D 295,1993-12-12,WXYZ

D 321,2021-06-22,MNOP

;

run;



* Code SAS souhaité pour traiter le fichier texte;

data TabOut;

   infile inVD dsd truncover fistobs=7;

   input @3

         NMVAR : 3.

         DTVAR : YYMMDD10.

         CHVAR : $CHAR4.

         ;

 run;





* Approche par génération de code ;



data _null_ ;

  length VarLu $ 32 FmtLu $ 32 FmtStr $ 200;

  retain FmtStr;



  * Première partie du code à générer; 

  if _n_=1 then do; 

           call execute("data TabOut;");

           call execute("  infile inVD dsd truncover;");

           call execute("  input @3 ");

           end;



  infile inVD;

  input RecType $upcase1. @;



  * Traiter les définitions de variables ;

  if RecType = "V" then

    do;

      input VarLu $ FmtLu $;

      call execute("        "|| catx(" ", VarLu, ":", FmtLu));

    end;

  else



  * Trouver la première ligne de données ;

  if RecType = "D" then

    do;

      call execute("        ;");

      call execute("  infile inVD  firstobs="||put(_n_,z3.)||";");

      call execute("run;");

      stop; 

    end;

run ;



*---------------------------------------;

* CALL EXECUTE dans la procédure REPORT ;

*---------------------------------------;



data Eleves;

  input  Nom :$8. Genre :$1. Age;

datalines;

Claude   M 17

Elodie   F 13

Samuel   X 18

Francois M 12

Phong    F 14

Tiago    M 15

Mila     F 16

;

run;



title1 "Table Eleves";

proc print data=Eleves;

run;



* Macro pour écrire dasn le journal les valeurs des variables de la table ;

%macro EcrireValeurs();

   call execute(

        cat('%put INFO: >          '

             , '_BREAK=' , substr(_break_, 1, 10)

             , 'Genre='  , Genre   , ' '

             , 'Nom='    , Nom     , ' '

             , 'Age='    , Age     , ' '

             , 'AgeMoy=' , AgeMoy  , ' '

             , ';'

           ));

%mend EcrireValeurs;



title1 "Rapport de la procédure";

proc report data=Eleves missing /* list showall */  out=RptEleves;

  columns Genre Nom Age Age=AgeMoy;

  define  Genre  / order;

  define  Nom    / display;

  define  Age    / display;

  define  AgeMoy / mean;

 

  compute after;

    call execute('%put NOTE: Compute After;');

    %EcrireValeurs();

  endcomp;

 

  compute after Genre;

    call execute('%put NOTE: Compute After Genre;');

    %EcrireValeurs();

  endcomp;



  compute Age;   

    call execute('%put NOTE: Compute Age;');

    %EcrireValeurs();

  endcomp;

run;



title1 "Fichier créé par la procédure";

proc print;

run;



*----------------------------------------------------------------------------------;

*  Exemple de génération de code avec une combinaison d'étape DATA et du %INCLUDE: ;

*----------------------------------------------------------------------------------;



* Liste des tables choisies ;

data TablesChoisies;

  input NomTable $ 32.;

datalines;

class

cars

heart

;

run;



* Fichier temporaire pour accueillir le code SAS généré ;

filename CodeSAS temp;



* Génération du code SAS;

data _null_;

  set TablesChoisies;

  file CodeSAS;

  put "data "  NomTable  ";" ;

  put "  set sashelp." NomTable ";";

  put "run;";

run;



%include CodeSAS /source2;





* Arrêt du traitement si table vide ;

* Version macro en "open code" ;



%let dset=TableVide;

%let dsid = %sysfunc(open(&dset));

%let nobs = %sysfunc(attrn(&dsid,NOBS));

%if &nobs=0 %then %do;

      %let rc = %sysfunc(close(&dsid));

      %put NOTE: &=dset  &=dsid  &=nobs  &=rc;

      endsas;

    %end;

    

*---------------------------------------------------;

* Clin d'oeil à la fonction DOSUBL VS CALL EXECUTE: ;

*---------------------------------------------------;



data _null_;

  putlog "AAA";

  call execute('data _null_;   ');

  call execute('  putlog "BBB";');

  call execute('run;           ');

  putlog "CCC";

run;



data _null_;

  putlog "AAA";

  rc = dosubl('data _null_;'    ||

              '  putlog "BBB";' ||

              'run;');

  putlog "CCC";

run;



*---------------------------------------------------;

* Accès répété à la même source de données:         ;

* Approche à éviter pour des raisons de performance ;

*---------------------------------------------------;



data _null_;

   set SASHELP.CARS;

    by MAKE;

   if first.MAKE then

      do;

        call execute('title1 "Manufacturier '||trim(MAKE)||'";');

        call execute('proc print noobs data=SASHELP.CARS(where=(MAKE="'||trim(MAKE)||'"));');

        call execute('  var MODEL TYPE;');

        call execute('run;');

     end;

run;

